Multivariate Within-Person Fluctuation across Staff-Rated Affect & Self-Reported Mood

Source: Kolanowski, A., **Hoffman, L.**, & Hofer, S. M. (2007). Concordance of self-report and informant assessment of emotional well-being in nursing home residents with dementia. *Journal of Gerontology: Psychological Sciences, 62B*(1), P20-27. Available at: http://digitalcommons.unl.edu/psychfacpub/420/.

SAS Data Set-Up:

```
* Re-stacking data into 4 rows per time point per person;
DATA Stack4; SET &datafile.;
      * DV variable gets used as categorical in REPEATED statement;
      * Other DV variables identify separate outcomes in MODEL and RANDOM;
      y=posaff;  dv=1; dvpa=1; dvna=0; dvpm=0; dvnm=0; OUTPUT;
      y=negaff;  dv=2; dvpa=0; dvna=1; dvpm=0; dvnm=0; OUTPUT;
      y=posmood; dv=3; dvpa=0; dvna=0; dvpm=1; dvnm=0; OUTPUT;
      y=negmood; dv=4; dvpa=0; dvna=0; dvpm=0; dvnm=1; OUTPUT; run;
* Figuring out which nearest case for printing R matrix;
DATA &udatafile.; SET &udatafile.;
    IF NMISS(posaff, negaff, posmood, negmood)=0 THEN Complete=1; ELSE Complete=0; run;
PROC PRINT DATA=&udatafile.; WHERE Complete=1; VAR ID Day Complete; run;
```

Obs	ID	Day	Complete
1	1	1	1
2	1	2	1
3	1	3	1

...

```
TITLE1 "Multivariate Model of Pos/Neg Affect/Mood";
TITLE2 "Printing R Matrix for Case #1 with R=1 and RCORR=1";
PROC MIXED DATA=Stack4 NOCLPRINT NOITPRINT COVTEST MAXIT=1000 METHOD=REML;
      CLASS ID Day DV;
      MODEL y= dvpa dvna dvpm dvnm / NOINT SOLUTION DDFM=Satterthwaite;
      RANDOM   dvpa dvna dvpm dvnm / G GCORR TYPE=UN SUBJECT=ID;  * Level 2 persons;
      REPEATED DV / R=1 RCORR=1 TYPE=UN SUBJECT=Day*ID;     * Level 1 crossed time*DV;
run;
```

Dimensions	
Covariance Parameters	20
Columns in X	4
Columns in Z Per Subject	4
Subjects	31 # persons
Max Obs Per Subject	48 # occasions * 4 DVs

Estimated R Matrix for ID*Day 1 1

Row	Col1	Col2	Col3	Col4
1	0.2074	-0.1340	-0.00963	0.006026
2	-0.1340	0.2896	-0.01422	0.01786
3	-0.00963	-0.01422	0.3071	-0.1281
4	0.006026	0.01786	-0.1281	0.2543

Estimated R Correlation Matrix for ID*Day 1 1

Row	Col1	Col2	Col3	Col4
1	1.0000	-0.5469	-0.03816	0.02624
2	**-0.5469**	1.0000	-0.04768	0.06581
3	**-0.03816**	-0.04768	1.0000	-0.4584
4	0.02624	**0.06581**	**-0.4584**	1.0000

Estimated G Matrix

Row	Effect	Subject identification	Col1	Col2	Col3	Col4
1	dvpa	1	0.3196	-0.1635	0.07596	-0.01172
2	dvna	1	-0.1635	0.1887	-0.06060	0.05021
3	dvpm	1	0.07596	-0.06060	0.2142	-0.1337
4	dvnm	1	-0.01172	0.05021	-0.1337	0.2162

Estimated G Correlation Matrix

Row	Effect	Subject identification	Col1	Col2	Col3	Col4
1	dvpa	1	1.0000	-0.6656	0.2903	-0.04461
2	dvna	1	**-0.6656**	1.0000	-0.3014	0.2486
3	dvpm	1	**0.2903**	-0.3014	1.0000	-0.6214
4	dvnm	1	-0.04461	**0.2486**	**-0.6214**	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	ID	0.3196	0.08733	3.66	0.0001	
UN(2,1)	**ID**	**-0.1635**	**0.05871**	**-2.78**	**0.0054**	**BP Covariance of Pos Affect & Neg Affect**
UN(2,2)	ID	0.1887	0.05542	3.41	0.0003	
UN(3,1)	**ID**	**0.07596**	**0.05400**	**1.41**	**0.1595**	**BP Covariance of Pos Affect & Pos Mood**
UN(3,2)	ID	-0.06060	0.04313	-1.41	0.1600	
UN(3,3)	ID	0.2142	0.06250	3.43	0.0003	
UN(4,1)	ID	-0.01172	0.05195	-0.23	0.8215	
UN(4,2)	**ID**	**0.05021**	**0.04243**	**1.18**	**0.2367**	**BP Covariance of Neg Affect & Neg Mood**
UN(4,3)	**ID**	**-0.1337**	**0.05126**	**-2.61**	**0.0091**	**BP Covariance of Pos Mood & Neg Mood**
UN(4,4)	ID	0.2162	0.06173	3.50	0.0002	
UN(1,1)	ID*Day	0.2074	0.01658	12.51	<.0001	
UN(2,1)	**ID*Day**	**-0.1340**	**0.01578**	**-8.49**	**<.0001**	**WP Covariance of Pos Affect & Neg Affect**
UN(2,2)	ID*Day	0.2896	0.02313	12.52	<.0001	
UN(3,1)	**ID*Day**	**-0.00963**	**0.01428**	**-0.67**	**0.4999**	**WP Covariance of Pos Affect & Pos Mood**
UN(3,2)	ID*Day	-0.01422	0.01687	-0.84	0.3993	
UN(3,3)	ID*Day	0.3071	0.02455	12.51	<.0001	
UN(4,1)	ID*Day	0.006026	0.01298	0.46	0.6426	
UN(4,2)	**ID*Day**	**0.01786**	**0.01536**	**1.16**	**0.2451**	**WP Covariance of Neg Affect & Neg Mood**
UN(4,3)	**ID*Day**	**-0.1281**	**0.01737**	**-7.37**	**<.0001**	**WP Covariance of Pos Mood and Neg Mood**
UN(4,4)	ID*Day	0.2543	0.02032	12.51	<.0001	

Fit Statistics

-2 Res Log Likelihood	2137.3
AIC (smaller is better)	2177.3
AICC (smaller is better)	2178.0
BIC (smaller is better)	2206.0

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
dvpa	-0.3049	0.1046	30.1	-2.92	0.0066	Intercept for Rated Positive Affect
dvna	0.3030	0.08339	30.3	3.63	0.0010	Intercept for Rated Negative Affect
dvpm	-0.06084	0.08849	30.2	-0.69	0.4970	Intercept for Self-Reported Positive Mood
dvnm	0.02001	0.08795	30.2	0.23	0.8216	Intercept for Self-Reported Negative Mood

From the paper (p. P24):

Between-person (BP) and within-person (WP) relations of rated and self-reported emotional well-being. We estimated empty multivariate multilevel models in order to examine the between-person and within-person correlations simultaneously among informant-rated and self-reported positive and negative emotion. There were significant negative correlations between informant-rated positive and negative emotion (BP $r = -.66$, $p < .01$; WP $r = -.55$, $p < .01$), and between self-reported positive and negative emotion (BP $r = -.61$, $p < .01$; WP $r = -.53$, $p < .01$). This suggests that raters and residents were each internally consistent at the between-person, individual level, as well as at the within-person, day level. For example, residents who were rated or self-reported high in overall positive emotion (relative to the rest of the sample) were also rated or self-reported low in overall negative emotion (relative to the rest of the sample). Similarly, on days where a resident was high on positive emotion (relative to him- or herself), he or she was also low in negative emotion (relative to himself or herself).

Correspondence between informant ratings and self-reported responses within each of the positive and negative dimensions of emotion was not as strong, however. Informant-rated and self-reported positive emotion were not significantly correlated (BP $r = .30$, $p = .13$; WP $r = .05$, $p = .37$), indicating overall levels (between persons) and daily levels (within persons) of positive emotion were not related across sources. Informant-rated and self-reported negative emotion were not significantly correlated between persons (BP $r = .30$, $p = .15$) but were significantly correlated within persons (WP $r = .17$, $p < .01$), indicating that although overall levels of negative emotion were not related between persons, daily levels of negative emotion were related within persons. Although in the current sample of 31 persons the statistical power to detect a between-person correlation of .30 was less than .50, a correlation of .30 would not indicate acceptable reliability, regardless of statistical significance.